$x^5 - 10x^3 + 5x^2 + 10x + 1$ の厳密解

with *Mathematica* 13.3

Galois群がC_5 のとき　　2025 年2月 by mixedmoss

§0. [準備] 原始元，分解方程式，ガロア群

以下は GaloisGroupProgram.nb を使って求めた結果です．

In[1345]:=

```
ClearAll["`*"];
f = x^5 - 10 x^3 + 5 x^2 + 10 x + 1;
```

f =0の解を x1, x2, x3, x4, x5 . f(x)=0 の原始元の一つをv = x1+2 x2 + 3 x3 + 4 x4 + 5 x5.　vの最小多項式を V[x] とすると

In[1347]:=

```
V[x_] = -4375 + 2500 x - 125 x^3 + x^5;
```

また x1, x2, x3, x4, x5 を原始元vで表すと

In[1348]:=

$$\{x1, x2, x3, x4, x5\} = \left\{\frac{35250 - 5125 v - 2575 v^2 + 25 v^3 + 22 v^4}{5375}, \frac{-98500 + 16875 v + 7325 v^2 - 140 v^3 - 63 v^4}{5375}, \frac{41500 - 10875 v - 3550 v^2 + 95 v^3 + 32 v^4}{5375}, \frac{71500 - 13750 v - 4575 v^2 + 130 v^3 + 37 v^4}{5375}, \frac{-49750 + 12875 v + 3375 v^2 - 110 v^3 - 28 v^4}{5375}\right\};$$

f(x) のGalois群を Gal , σ=(1,2,3,4,5) とすると

$\mathrm{Gal} = \{Id, \sigma, \sigma^2, \sigma^3, \sigma^4\} = C_5$

ただし，C_5 は5次巡回群で，分解列は $\mathrm{Gal} = C_5 \triangleright \{Id\}$

§1．Lagrange分解式 r1,r2,r3,r4 の5乗の値を求める

$C_5 \triangleright$ {Id}の部分に注目し，ラグランジュの5次分解式をつくる.(ζは1の原始 5 乗根のひとつ．)
このとき，V(v)と円分方程式 $v^4+v^3+v^2+v+1$ で割った余りを考える．(PolynomialMod)

In[1349]:=
```
r0 = x1 + x2 +  x3 + x4 + x5 // PolynomialMod[#, V[v]] &;
r1 =
  x1 + ξ x2 + ξ^2 x3 + ξ^3 x4 + ξ^4 x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
r2 = x1 + ξ^2 x2 + ξ^4 x3 + ξ x4 + ξ^3 x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
r3 = x1 + ξ^3 x2 + ξ x3 + ξ^4 x4 + ξ^2 x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
r4 = x1 + ξ^4 x2 + ξ^3 x3 + ξ^2 x4 + ξ x5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &;
```

解と係数の関係より r0 = 0. またσにより分解式は次の様に移る.

$$
(\#)\begin{cases}
r1 \to \zeta^4\, r1 \to \zeta^3\, r1 \to \zeta^2\, r1 \to \zeta\, r1 \to r1 \quad \square \\
r2 \to \zeta^3\, r2 \to \zeta\, r2 \to \zeta^4\, r2 \to \zeta^2\, r2 \to r2 \\
r3 \to \zeta^2\, r3 \to \zeta^4\, r3 \to \zeta\, r3 \to \zeta^3\, r3 \to r3 \\
r4 \to \zeta\, r4 \to \zeta^2\, r4 \to \zeta^3\, r4 \to \zeta^4\, r4 \to r4
\end{cases}
$$

さらに 同型写像 $\omega: \zeta \to \zeta^3$ とすると，分解式は次のように移る.

(by ω) r1→r3→r4→r2

R_1=r1^5, R_2 = r2^5, R_3=r3^5, R_4 = r4^{5}とおく．(#)より R_1,R_2,R_3,R_4は σで不変なのでC_5 で不変．故に Q(ζ)に入るはず．「単純計算」して

In[1354]:=
```
r1^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
r2^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
r3^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
r4^5 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
```

Out[1354]=
$3125\,\xi^3$

Out[1355]=
$3125\,\xi$

Out[1356]=
$-3125 - 3125\,\xi - 3125\,\xi^2 - 3125\,\xi^3$

Out[1357]=
$3125\,\xi^2$

Mathematica は何故か直してないが，$1+\xi+\xi^2+\xi^3+\xi^4 = 0$ だから，$(r3)^5 = 3125\,\xi^4 = 5^5\,\xi^4$
故に $(R_1, R_2, R_3, R_4) = \left(5^5\,\xi^3, 5^5\,\xi, 5^5\,\xi^4, 5^5\,\xi^2\right)$

あとは5乗根を取れば良いだけだが，偏角の取り方が 5 通りもあるから，これからが難しい.

(実は「$R_1 = 5^5\,\xi^3$」を ωで移せば，残りの 3 つもすぐ求まる．)

§2. r1 r2^2, r2 r4^2, r1^{2}r3, r2 r4^2 の値を考慮しr1,r2,r3,r4 の値を求める

2 - 1. r1 r4, r2 r3, r1^{2}r3, r1 r2^2 の値を求める

$$(\#\#)\begin{cases} x1 &= 1/5\,(r1 + r2 + r3 + r4) \\ x2 &= 1/5\left(\zeta^4\, r1 + \zeta^3\, r2 + \zeta^2\, r3 + \zeta\, r4\right) \\ x3 &= 1/5\left(\zeta^3\, r1 + \zeta\, r2 + \zeta^4\, r3 + \zeta^2\, r4\right) \\ x4 &= 1/5\left(\zeta^2\, r1 + \zeta^4\, r2 + \zeta\, r3 + \zeta^3\, r4\right) \\ x5 &= 1/5\left(\zeta\, r1 + \zeta^2\, r2 + \zeta^3\, r3 + \zeta^4\, r4\right) \end{cases}$$

まず ζを $\zeta^2, \zeta^3, \zeta^4$ に変えても 解集合{x1,x2,x3,x4,x5}は変わらない.

§1(#)より, r1r4, r2r3, r1 r2^2,r1^{2}r3は σによって不変なので , C_5の固定体Q(ζ)に入る. 「単純計算」して,

In[1358]:=

```
r1 r4 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
r2 r3 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
r1 r2^2 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
r1 ^2 r3 // PolynomialMod[#, {V[v], Cyclotomic[5, ξ]}] &
```

Out[1358]=

25

Out[1359]=

25

Out[1360]=

$-125 - 125\,\xi - 125\,\xi^2 - 125\,\xi^3$

Out[1361]=

$125\,\xi^2$

「$1 + \xi + \xi^2 + \xi^3 + \xi^4 = 0$」だから, 上の結果より,

$$\begin{cases} r1\ r4 = 25 & \cdots (1) \\ r2\ r3 = 25 & \cdots (2) \\ r1\ r2^2 = 125\,\xi^4 & \cdots (3) \\ r3\ r1^2 = 125\,\xi^2 & \cdots (4) \end{cases}$$

(実は「r1 r4 = 25 と r1 r2^2 = 125 ξ^4」を ωで移せば, 残りの2つもすぐ求まる.)

(A)三角関数を利用してr1,r2,r3,r4を求める

r1とr4, r2とr3は互いに複素共役だから(「補充」参照), (1)(2)より

r1= 5(cos θ_1+ i sin θ_1) , r2= 5(cos θ_2+ i sin θ_2), r3= 5(cos(- θ_2)+ i sin(-θ_2)), r4= 5(cos(- θ_1)+ i sin(-θ_1))

とおける. ここで「ζ=cos($\frac{2\pi}{5}$)+i sin($\frac{2\pi}{5}$)」としても一般性を失わない. このとき(3) と(4)より

θ_1+2θ_2=$\frac{8\pi}{5}$ かつ 2θ_1-θ_2=$\frac{4\pi}{5}$

∴ θ_1=$\frac{16\pi}{25}$ かつ θ_2=$\frac{12\pi}{25}$

即ち (A) $\begin{cases} r1 & = 5\,(\cos(16\pi/25) + i\sin(16\pi/25)) \\ r2 & = 5\,(\cos(12\pi/25) + i\sin(12\pi/25)) \\ r3 & = 5\,(\cos(-12\pi/25) + i\sin(-12\pi/25)) \\ r4 & = 5\,(\cos(-16\pi/25) + i\sin(-16\pi/25)) \end{cases}$

(B)累乗根を利用して r1,r2,r3,r4を求める(その1)

複素数$z = r(\cos\theta + i\sin\theta)$ $(r > 0,\ -\pi < \theta < \pi)$ に対し,

$z^{\frac{1}{5}} = r^{\frac{1}{5}}\left(\cos\left(\frac{\theta}{5}\right) + i\cos\left(\frac{\theta}{5}\right)\right)$ と定義する. (Mathematicaも多分同じ定義)

それに対し $\sqrt[5]{z} = r^{\frac{\theta}{5}}\left(\cos\left(\frac{\theta}{5} + \frac{2k\pi}{5}\right) + i\cos\left(\frac{\theta}{5} + \frac{2k\pi}{5}\right)\right)$ (kは適当な整数) と定義すると,

いずれも5 乗すればzであるが, $\sqrt[5]{z}$ の方は偏角が5通りある.(この定義はここだけの話です)

このとき (1) より$(r1, r2, r3, r4) = \left(5\sqrt[5]{\zeta^3},\ 5\sqrt[5]{\zeta},\ 5\sqrt[5]{\zeta^4},\ 5\sqrt[5]{\zeta^2}\right)$

(A)を同値変形してみる. 同値変形なので, ζの偏角を $\frac{2\pi}{5}$ とする. $z^{\frac{1}{5}}$において zの偏角は $(-\pi)$から πで考えているから,

$$\arg((\zeta^3)^{\frac{1}{5}}) = \frac{1}{5}\left(-\frac{4\pi}{5}\right) = -\frac{4\pi}{25}$$

一方「$\arg(r_1) = \theta_1 = \frac{16\pi}{25}$ 」だから, $\sqrt[5]{\zeta^3} = (\zeta^3)^{\frac{1}{5}} \times \zeta^2$

同様にして $\sqrt[5]{\zeta} = (\zeta)^{\frac{1}{5}} \times \zeta$, $\sqrt[5]{\zeta^4} = (\zeta^4)^{\frac{1}{5}} \times \zeta^4$, $\sqrt[5]{\zeta^2} = (\zeta^2)^{\frac{1}{5}} \times \zeta^3$

故に, (B) $\begin{cases} r1 & = 5\sqrt[5]{\zeta^3} = 5\,(\zeta^3)^{\frac{1}{5}} \times \zeta^2 \\ r2 & = 5\sqrt[5]{\zeta} = 5\,(\zeta)^{\frac{1}{5}} \times \zeta \\ r3 & = 5\sqrt[5]{\zeta^4} = 5\,(\zeta^4)^{\frac{1}{5}} \times \zeta^4 \\ r4 & = 5\sqrt[5]{\zeta^2} = 5\,(\zeta^2)^{\frac{1}{5}} \times \zeta^3 \end{cases}$

(C)累乗根を利用して r1,r2,r3,r4を求める(その2)

今度は「Galois群が $D_5 \wedge R_i$が虚数」の時も考え「(A)は無かったものとして」最初から解いてみる．(3)(4)を満たす複素数解を求めるには，参考文献「可解な5次方程式について」の補題が役立つ．それは「r_1は R_1の任意の5乗根にとっても一般性を失わない」という補題である．すなわち「r_1を R_1の特定の5乗根にとって(3)(4)が解を持つ時，r_1を R_1の別の5乗根にとっても(3)(4)は解を持ち，fの解の集合は同じとなる」ということ．

さて，$\zeta=\cos(\frac{2\pi}{5})+i\sin(\frac{2\pi}{5})$ としてみる．そして $r_1=5\sqrt[5]{\zeta^3}=5\left(\zeta^3\right)^{\frac{1}{5}}$ と（勝手に）決める．このとき

$$r_2=\{5(\zeta)^{\frac{1}{5}},5(\zeta)^{\frac{6}{5}},5(\zeta)^{\frac{11}{5}},5(\zeta)^{\frac{16}{5}},5(\zeta)^{\frac{21}{5}}\}$$

の何れかとなる．これを(3)(4)へ代入して適するのを見つけると，$r_2=5(\zeta)^{\frac{11}{5}}$であることが分かる．
r1とr4, r2とr3は互いに複素共役だから

$$(C)\begin{cases} r1 & =5\sqrt[5]{\zeta^3} & =5\left(\zeta^3\right)^{\frac{1}{5}} \\ r2 & =5\sqrt[5]{\zeta} & =5(\zeta)^{\frac{1}{5}}\times\zeta^2 \\ r3 & =5\sqrt[5]{\zeta^4} & =5\left(\zeta^4\right)^{\frac{1}{5}}\times\zeta^3 \\ r4 & =5\sqrt[5]{\zeta^2} & =5\left(\zeta^2\right)^{\frac{1}{5}} \end{cases}$$

(B) と (C) は集合として異なるが，同じfの解を与える．

§3. x1,x2,x3,x4,x5 の値を求める

3通りの方法で求めて，それを比較する.

3-1. (A) を利用して， x1, x2, x3, x4, x5 を三角関数で求める

In[1362]:=
```
ξ = Cos[2 Pi / 5] + I Sin[2 Pi / 5] // Simplify;
```

r1,r2,..の値を r1A, r2A, .. とする.

In[1363]:=
```
θ1 = 16 Pi / 25; θ2 = 12 Pi / 25;
r1A = 5 (Cos[θ1] + I Sin[θ1]) // Simplify;
r2A = 5 (Cos[θ2] + I Sin[θ2]) // Simplify;
r3A = 5 (Cos[-θ2] + I Sin[-θ2]) // Simplify;
r4A = 5 (Cos[-θ1] + I Sin[-θ1]) // Simplify;
```

三角関数で表されたf (x) の解を x1A, x2A, ... とすると,

In[1368]:=
```
x1A = 1 / 5 (r1A + r2A + r3A + r4A) // Simplify
x2A = 1 / 5 (ξ^4 r1A + ξ^3 r2A + ξ^2 r3A + ξ r4A) // Simplify
x3A = 1 / 5 (ξ^3 r1A + ξ r2A + ξ^4 r3A + ξ^2 r4A) // Simplify
x4A = 1 / 5 (ξ^2 r1A + ξ^4 r2A + ξ r3A + ξ^3 r4A) // Simplify
x5A = 1 / 5 (ξ r1A + ξ^2 r2A + ξ^3 r3A + ξ^4 r4A) // Simplify
```

Out[1368]=
$$2\left(\mathrm{Sin}\left[\frac{\pi}{50}\right]-\mathrm{Sin}\left[\frac{7\pi}{50}\right]\right)$$

Out[1369]=
$$\frac{1}{4}\left(\sqrt{10\left(5+\sqrt{5}\right)}\,\mathrm{Cos}\left[\frac{\pi}{50}\right]+\sqrt{2\left(5+\sqrt{5}\right)}\left(-\mathrm{Cos}\left[\frac{\pi}{50}\right]+2\,\mathrm{Cos}\left[\frac{7\pi}{50}\right]\right)-2\left(\left(1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{\pi}{50}\right]+\left(-1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{7\pi}{50}\right]\right)\right)$$

Out[1370]=
$$\frac{1}{4}\left(\sqrt{10\left(5+\sqrt{5}\right)}\,\mathrm{Cos}\left[\frac{7\pi}{50}\right]-\sqrt{2\left(5+\sqrt{5}\right)}\left(2\,\mathrm{Cos}\left[\frac{\pi}{50}\right]+\mathrm{Cos}\left[\frac{7\pi}{50}\right]\right)+2\left(-1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{\pi}{50}\right]+2\left(1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{7\pi}{50}\right]\right)$$

Out[1371]=
$$\frac{1}{4}\left(-\sqrt{10\left(5+\sqrt{5}\right)}\,\mathrm{Cos}\left[\frac{7\pi}{50}\right]+\sqrt{2\left(5+\sqrt{5}\right)}\left(2\,\mathrm{Cos}\left[\frac{\pi}{50}\right]+\mathrm{Cos}\left[\frac{7\pi}{50}\right]\right)+2\left(-1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{\pi}{50}\right]+2\left(1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{7\pi}{50}\right]\right)$$

Out[1372]=
$$\frac{1}{4}\left(-\sqrt{10\left(5+\sqrt{5}\right)}\,\mathrm{Cos}\left[\frac{\pi}{50}\right]+\sqrt{2\left(5+\sqrt{5}\right)}\left(\mathrm{Cos}\left[\frac{\pi}{50}\right]-2\,\mathrm{Cos}\left[\frac{7\pi}{50}\right]\right)-2\left(\left(1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{\pi}{50}\right]+\left(-1+\sqrt{5}\right)\mathrm{Sin}\left[\frac{7\pi}{50}\right]\right)\right)$$

3-2. (B) を利用して，x1, x2, x3, x4, x5 を累乗根で求める

r1, r2, .. の値を r1B, r2B, .. とする.

In[1373]:=

```
r1B = 5 (ξ^3) ^ (1 / 5) * ξ^2 // Simplify;
r2B = 5 (ξ) ^ (1 / 5) * ξ // Simplify;
r3B = 5 (ξ^4) ^ (1 / 5) * ξ^4 // Simplify;
r4B = 5 (ξ^2) ^ (1 / 5) * ξ^3 // Simplify;
```

累乗根で表されたf (x) の解を x1B, x2B, ... とすると,

In[1377]:=

```
x1B = 1 / 5 (r1B + r2B + r3B + r4B) // Simplify;
x2B = 1 / 5 (ξ^4 r1B + ξ^3 r2B + ξ^2 r3B + ξ r4B) // Simplify;
x3B = 1 / 5 (ξ^3 r1B + ξ r2B + ξ^4 r3B + ξ^2 r4B) // Simplify;
x4B = 1 / 5 (ξ^2 r1B + ξ^4 r2B + ξ r3B + ξ^3 r4B) // Simplify;
x5B = 1 / 5 (ξ r1B + ξ^2 r2B + ξ^3 r3B + ξ^4 r4B) // Simplify;
x1B // TraditionalForm // Print
x2B // TraditionalForm // Print
x3B // TraditionalForm // Print
x4B // TraditionalForm // Print
x5B // TraditionalForm // Print
```

$$-\frac{1}{512\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{6/5}\left(-8+(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{7/5}\right)\left(16\times 2^{2/5}+\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{11/5}\right)$$

$$-\frac{1}{8192\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{21/5}\left(2^{2/5}+\sqrt[5]{-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}}\right)\left(-32+(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{12/5}\right)$$

$$-\frac{1}{131\,072\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{11/5}\left(64\times 2^{2/5}+\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{16/5}\right)\left(-128+(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{17/5}\right)$$

$$\frac{1}{4096\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{23/5}\left(-2\,(-1)^{3/5}\sqrt[5]{2}+\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{3/5}\right)\left(4\times 2^{2/5}+\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{6/5}\right)$$

$$-\frac{1}{32\,768\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{16/5}\left(-2+(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2/5}\right)\left(256\times 2^{2/5}+\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{21/5}\right)$$

3-3. (C) を利用して, x1, x2, x3, x4, x5 を累乗根で求める

r1, r2, .. の値を r1C, r2C, .. とする.

In[1387]:=

```
r1C = 5 (ξ^3) ^ (1 / 5) // Simplify;
r2C = 5 (ξ) ^ (1 / 5) * ξ^2 // Simplify;
r3C = 5 (ξ^4) ^ (1 / 5) * ξ^3 // Simplify;
r4C = 5 (ξ^2) ^ (1 / 5) // Simplify;
```

累乗根で表されたf (x) の解を x1C, x2C, ... とすると,

In[1391]:=

```
x1C = 1 / 5 (r1C + r2C + r3C + r4C) // Simplify;
x2C = 1 / 5 (ξ^4 r1C + ξ^3 r2C + ξ^2 r3C + ξ r4C) // Simplify;
x3C = 1 / 5 (ξ^3 r1C + ξ r2C + ξ^4 r3C + ξ^2 r4C) // Simplify;
x4C = 1 / 5 (ξ^2 r1C + ξ^4 r2C + ξ r3C + ξ^3 r4C) // Simplify;
x5C = 1 / 5 (ξ r1C + ξ^2 r2C + ξ^3 r3C + ξ^4 r4C) // Simplify;
x1C // TraditionalForm // Print
x2C // TraditionalForm // Print
x3C // TraditionalForm // Print
x4C // TraditionalForm // Print
x5C // TraditionalForm // Print
```

$$\frac{1}{128\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{2/5}\left(128-64\,(-2)^{3/5}\sqrt[5]{-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}}+8\times 2^{2/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{9/5}-(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{17/5}\right)$$

$$\frac{1}{2048\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{7/5}\left(512-4\,(-2)^{3/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{16/5}+2\times 2^{2/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{19/5}-(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{22/5}\right)$$

$$\frac{1}{32\,768\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{12/5}\left(2048+512\times 2^{2/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{4/5}-256\,(-2)^{3/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{6/5}-(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{27/5}\right)$$

$$\frac{1}{4096\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{13/5}\left(-128\,(-2)^{3/5}+64\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{4/5}-8\,(-1)^{3/5}\sqrt[5]{2}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{11/5}+2^{2/5}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{18/5}\right)$$

$$-\frac{1}{256\times 2^{4/5}}\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{8/5}\left(2^{2/5}+\sqrt[5]{-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}}\right)\left(32\,(-1)^{3/5}\sqrt[5]{2}-\left(-1+\sqrt{5}+i\sqrt{2\left(5+\sqrt{5}\right)}\right)^{13/5}\right)$$

(B)(C)での5乗根は, 虚数の5乗根であることに注意されたい.

3-3.「三角関数で表された解(A)」と「累乗根で表された解(B)&(C)」を比較する

厳密な比較は難しいので近似値を求めると

In[1401]:=

```
N[{x1A, x2A, x3A, x4A, x5A}] // Sort
N[{x1B, x2B, x3B, x4B, x5B}] // Sort
N[{x1C, x2C, x3C, x4C, x5C}] // Sort
```

Out[1401]=

$\{-3.25908, -0.725978, -0.10694, 1.5624, 2.52959\}$

Out[1402]=

$\{-3.25908 - 1.11022\times10^{-15}\, i, -0.725978 + 2.42861\times10^{-16}\, i, -0.10694 - 4.16334\times10^{-17}\, i, 1.5624 - 1.9984\times10^{-15}\, i, 2.52959 + 2.22045\times10^{-16}\, i\}$

Out[1403]=

$\{-3.25908 + 1.21155\times10^{-15}\, i, -0.725978 - 2.39121\times10^{-16}\, i, -0.10694 + 1.99267\times10^{-16}\, i, 1.5624 - 1.91297\times10^{-16}\, i, 2.52959 + 6.66134\times10^{-16}\, i\}$

(B)(C)の方の解は，とても小さい虚部を持っている．見づらいので除くと，

In[1404]:=

```
trim[x_] := If[Abs@Im@N[x] < 10^(-14), Re@N[x], N[x]]
trim /@ {x1B, x2B, x3B, x4B, x5B} // Sort
trim /@ {x1C, x2C, x3C, x4C, x5C} // Sort
```

Out[1405]=

$\{-3.25908, -0.725978, -0.10694, 1.5624, 2.52959\}$

Out[1406]=

$\{-3.25908, -0.725978, -0.10694, 1.5624, 2.52959\}$

10^{-14} 未満の虚部を除くと3種類の解は一致するので，
この3つは一致しかつf の解は全て実数と見て良い．念の為グラフも描いてみる．

In[1407]:=

```
Plot[f, {x, -4, 3}, PlotLegends → Placed["y=f(x)=x^5-10x^3+5x^2+10x+1", {Bottom, Left}]]
```

Out[1407]=



$y=f(x)=x^5-10x^3+5x^2+10x+1$

§4. まとめ

ガロア群が C_5 の場合

[Step1] 5次のLagrangeの分解式を作り r_1, r_2, r_3, r_4 の5乗 R_1, R_2, R_3, R_4 の値を求める (ζ は 1 の原始 5 乗根の一つ)

$$\begin{cases} r_0 &= x_1 + x_2 + x_3 + x_4 + x_5\ (=0) \\ r_1 &= x_1 + \zeta\, x_2 + \zeta^2\, x_3 + \zeta^3\, x_4 + \zeta^4\, x_5 \\ r_2 &= x_1 + \zeta^2\, x_2 + \zeta^4\, x_3 + \zeta\, x_4 + \zeta^3\, x_5 \\ r_3 &= x_1 + \zeta^3\, x_2 + \zeta\, x_3 + \zeta^4\, x_4 + \zeta^2\, x_5 \\ r_4 &= x_1 + \zeta^4\, x_2 + \zeta^3\, x_3 + \zeta^2\, x_4 + \zeta\, x_5 \end{cases} \Rightarrow \begin{cases} x_1 &= r_1 + r_2 + r_3 + r_4 \\ x_2 &= \zeta^4\, r_1 + \zeta^3\, r_2 + \zeta^2\, r_3 + \zeta\, r_4 \\ x_3 &= \zeta^3\, r_1 + \zeta\, r_2 + \zeta^4\, r_3 + \zeta^2\, r_4 \\ x_4 &= \zeta^2\, r_1 + \zeta^4\, r_2 + \zeta\, r_3 + \zeta^3\, r_4 \\ x_5 &= \zeta\, r_1 + \zeta^2\, r_2 + \zeta^3\, r_3 + \zeta^4\, r_4 \end{cases}$$

ここで 次の関係が成り立つ.

$$(\#)\begin{cases} r_1 \text{ を}\sigma\text{で移すと } r_1 \to \zeta^4\, r_1 \quad \square \\ r_2 \text{ を}\sigma\text{で移すと } r_2 \to \zeta^3\, r_2 \\ r_3 \text{ を}\sigma\text{で移すと } r_3 \to \zeta^2\, r_3 \\ r_4 \text{ を}\sigma\text{で移すと } r_4 \to \zeta\, r_4 \end{cases}$$

σは 分解式を「 r1 → ζ^4 r1→ → ζ^3 r1→ → ζ^2 r1 → ζ r1 → r1」のように移す.

さらに $\omega: \zeta \to \zeta^3$ とすると, 分解式は次のように移る.

(by ω) r1→r3→r4→r2

(#)からR_1, R_2, R_3, R_4 は C_5 で不変. 故に $Q(\zeta)$ に入るので R_1, R_2, R_3, R_4 の値は「単純計算」で求まる. (方程式などは不要)

[Step2] R_k (1≦k≦4) の5乗根 r_k (1≦k≦4)のうち, $r_1^2\, r_3$, $r_2^2\, r_1$ の値を満たす r_k を求める. それを代入すれば f(x)の解が求まる

$r_1\, r_4$, $r_2\, r_3$, $r_1^2\, r_3$, $r_2^2\, r_1$ も $Q(\zeta)$ に入るので 単純計算で値は求まる. さらに r_1 と r_4, r_2 とr_3 が複素共役が言える (補足参照)

今の例ではうまく$r_1\, r_2^2$と$r_3\, r_1^2$の偏角が πの有理数倍になったので 角度がπの有理数倍の三角関数で表すことが出来た. しかしこれは一般には成り立たない. その場合は(A)の方法ではなく, (C)の方法で求めるしかない. すなわち R_1の 5乗根 r_1を1つ固定し,
R_2 の5乗根 r_2の5通りを試す. このようにして$r_1^2\, r_3$, $r_2^2\, r_1$を満たすr_1~r_4を見つけることができる.

【注】実際に必要な「単純計算」は,
本当は「R_1, $r_1\, r_4$, $r_1\, r_2^2$」の3つだけ. 後の式はこれを ω で移せば得られる. [以上]

§5. 補足

[補足1] r_1 と r_4, r_2 と r_3 は複素共役

$r_1{}^5 \in \mathrm{Q}(\zeta)$ だから, $r_1{}^5 = a + b\,\zeta + c\,\zeta^2 + d\,\zeta^3$ (a, b, c, d は有理数) とおける. これを $\omega^3 : \zeta \to \zeta^2$ で移すと

$$\begin{cases} r_1{}^5 = a + b\,\zeta^2 + c\,\zeta^4 + d\,\zeta & \square \\ r_2{}^5 = a + b\,\zeta^4 + c\,\zeta^3 + d\,\zeta^2 & \square \\ r_4{}^5 = a + b\,\zeta^3 + c\,\zeta + d\,\zeta^4 & \square \\ r_3{}^5 = a + b\,\zeta + c\,\zeta^2 + d\,\zeta^3 & \square \end{cases}$$

a, b, c, d は有理数だから r_1 と r_4, r_2 と r_3 は複素共役

[補足2] f の解は全て実数.

補足 1 より明らか.

[補足3]「解は　角度がπの有理数倍の三角関数で表される」とは限らない.

この例ではうまく「角度がπの有理数倍」の三角関数で表すことが出来たので, 私は結構長いこと, これはC_5のGalois群を持てば成り立つと予想していた. しかしC_5のGalois群を持つ他の方程式を解くとそうなっていないので, これは誤りと分かった. それどころか, 私は5個ほどC_5のGalois群を持つ方程式を解いたのだが, どれも「角度がπの有理数倍の三角関数」では表せなかった. どうやらこの例はとても特別な例らしい. この例は「モニックで, 4次の係数が0, 係数は整数で絶対値が10以下」という条件では, 唯一のGalois群が C_5 の方程式なのだが, この様な特別な性質を持っていたことは, とても幸運でした. 解いていてとても面白かった.

In[1408]:=

他の5次方程式は `https://mixedmoss/com/Mathematica/Galois` にお越しください